use "${clean}newspaper_analysis.dta", clear

// recoding Alexandria, VA to the South (technically part of DC until 1844)
replace south = 1 if city == "Alexandria"
replace north = 0 if city == "Alexandria"
replace state = "Virginia" if city == "Alexandria"

// dropping DC as a border area
drop if state == "District Of Columbia"

// keeping only the north
keep if north == 1

// standardizing variables
egen county_surban = std(county_urban_percent)
egen county_srugged = std(county_ruggedness)
egen county_spostoffice = std(county_postoffice_area)

// keeping the sample consistent
cap drop sample
gen sample = 1
foreach var of varlist $county_npcontrols {
	replace sample = . if `var' == .
}

// creating some variables
local c = 1860

gen period = year > `c'
replace year = year - `c'

encode state, gen(state_id)


* regressions
* ------------
estimates clear

// mcclellan baseline, city fes
xtset city_id
eststo: xtreg singular c.year##i.period##i.lincoln_county ${county_npcontrols} if sample == 1, fe cluster(city_id)
estadd local fe "City"

// mcclellan baseline, state fes
xtset state_id
eststo: xtreg singular c.year##i.period##i.lincoln_county ${county_npcontrols} if sample == 1, fe cluster(state_id)
estadd local fe "State"


// esttab
esttab est* using "${output}TabD12_np_lincoln_slopes_altfes_fulltable.tex", replace ///
	b(%8.3f) se(%8.3f) ///
	starlevels(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
	scalars("fe Fixed Effects") ///
	keep( ///
		year ///
		1.period ///
		1.lincoln_county ///
		1.period#c.year ///
		1.lincoln_county#c.year ///
		1.period#1.lincoln_county ///
		1.period#1.lincoln_county#c.year ///
		county_surban county_spostoffice county_srugged county_anyriver county_anycanal _cons ///
		) ///
	order( ///
		year ///
		1.period ///
		1.lincoln_county ///
		1.period#c.year ///
		1.lincoln_county#c.year ///
		1.period#1.lincoln_county ///
		1.period#1.lincoln_county#c.year ///
		county_surban county_spostoffice county_srugged county_anyriver county_anycanal _cons ///
		) ///
	coeflabels( ///
		year "Year" ///
		1.period "After 1860" ///
		1.lincoln_county "Lincoln County" ///
		1.period#c.year "Year $\times$ After 1860" ///
		1.lincoln_county#c.year "Year $\times$ Lincoln County" ///
		1.period#1.lincoln_county "After 1860 $\times$ Lincoln County" ///
		1.period#1.lincoln_county#c.year "Year $\times$ After 1860 $\times$ Lincoln County" ///
		county_surban "County \% Urban (std)" ///
		county_spostoffice "Post Office Density (std)" ///
		county_srugged "Terrain Ruggedness (std)" ///
		county_anyriver "County on River" ///
		county_anycanal "County on Canal" ///
		_cons "Constant" ///
		) ///
	nodepvars nomtitles nonotes ///
	nogaps nolines posthead(\midrule) prefoot(\midrule) postfoot(\bottomrule \end{tabular}}) ///
	tex compress	
	

	
	
	







